
Overview

This document lists a TTCN change needed to correct a problem in ATS ‘iwd-EUTRA-B2011-03_D11wk27’.

Table of Contents

2Overview

Table of Contents
2
Corrections
2
Change 1 –  Simplification of postamble in TC 9.2.1.1.7
2
Change 2 –  Function f_UT_SwitchOnUE
3
Change 3 – Function f_EUTRA_RmvFbdnPLMN
4


Corrections 

Change 1 – Simplification of postamble in TC 9.2.1.1.7
	Function name
	f_TC_9_2_1_1_7_EUTRA

	Reason for change
	- Neighbour cells do no longer matter after the Test Body is completed.

- Function ‘f_EUTRA_RmvFbdnPLMN’ is unnecessarily complex and thus extends the overall duration of the TC. See change 3.
- When returning from f_EUTRA_RmvFbdnPLMN the UE has to attach to Cell A instead of Cell I.

	Summary of change
	- Non-used cells are switched off before entering the postamble. 
- Function ‘f_EUTRA_RmvFbdnPLMN’ is simplified. This makes it more stable and also reduces the execution duration to less than 5 minutes.

- Postamble function is called with ‘eutra_CellA’ as parameter.

	TTCN module
	NAS_Attach_Eonly.ttcn

	MCC160 Comment
	Accepted


Before change
	    fl_TC_9_2_1_1_7_Body();
    // Postamble
    f_EUTRA_TestBody_Set(false);
    v_CellPowerList := {
      cs_CellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE ),
      cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),
      cs_CellPower ( eutra_CellA, tsc_NonSuitableCellRS_EPRE ),
      cs_CellPower ( eutra_CellG, tsc_NonSuitableCellRS_EPRE )
    };
    v_Guti_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellI); // @sic R5s100875 sic@
    f_EUTRA_RmvFbdnPLMN (eutra_CellA, eutra_CellI, v_CellPowerList, f_Asn2Nas_PlmnId(v_Guti_Params.PLMN_Identity), NORMAL); // @sic R5s110176, R5-106675 sic@
    f_EUTRA_Postamble(eutra_CellI, E1_IDLE, EPS_Only);  // @sic R5s110176, R5s110168 change 2.21 sic@
  } // function f_TC_9_2_1_1_7_EUTRA


After change

	    fl_TC_9_2_1_1_7_Body();
    // Postamble
    f_EUTRA_TestBody_Set(false);
    // Switch off cells not used
    f_EUTRA_SetCellPower ( eutra_CellJ, tsc_NonSuitableOffCellRS_EPRE );
    f_EUTRA_SetCellPower ( eutra_CellI, tsc_NonSuitableOffCellRS_EPRE );    
    f_EUTRA_SetCellPower ( eutra_CellG, tsc_NonSuitableOffCellRS_EPRE );    
    f_EUTRA_RmvFbdnPLMN (eutra_CellA, eutra_CellI, NORMAL); 
    f_EUTRA_Postamble(eutra_CellA, E1_IDLE, EPS_Only); 
  } // function f_TC_9_2_1_1_7_EUTRA


Change 2 – Extending ‘f_UT_SwitchOnUE’ with confirmation flag
	Function name
	f_UT_SwitchOnUE

	Reason for change
	In situations like the one described in change 3, the UE has to be switched on completely before sending additional AT commands. It is required to wait for the UE being completely switched on before continuing on TTCN level with either AT or MMI commands.

	Summary of change
	An additional optional flag (‘CNF_REQUIRED’ ) has been introduced to force TTCN waiting until the switch on command has been confirmed by the UE.

	TTCN module
	UpperTesterFunctions.ttcn

	MCC160 Comment
	Accepted


Before change

	  //----------------------------------------------------------------------------
  /*
   * @desc      switch on the UE
   * @param     p_Port
   * @param     p_USIM_Rmvd
   * @status    APPROVED
   */
  function f_UT_SwitchOnUE(UT_PTC_MTC_PORT p_Port,
                           boolean p_USIM_Rmvd)
  {
    var Ut_Common_Req_Type v_Request;
    if (p_USIM_Rmvd and pc_USIM_Rmv) {  // SIM needs to be inserted
      v_Request := INSERT_USIM;
    } else if (pc_SwitchOnOff) {        // switch on the UE
      v_Request := SWITCH_ON;
    } else if (pc_USIM_Rmv) {           // UE doesn't support switch off, so SIM needs to be inserted;  @sic R5-104083 sic@
      v_Request := INSERT_USIM;
    } else {                            // power on the UE
      v_Request := POWER_ON;
    }
    fl_UT_ApplyCommand(p_Port, cas_UT_Req(v_Request, NO_CNF_REQUIRED, omit));
  }


After change

	  //----------------------------------------------------------------------------
  /*
   * @desc      switch on the UE
   * @param     p_Port
   * @param     p_USIM_Rmvd
   * @status    APPROVED
   */
  function f_UT_SwitchOnUE(UT_PTC_MTC_PORT p_Port,
                           boolean p_USIM_Rmvd,
                           boolean p_CnfRequired := false)
  {
    var Ut_Common_Req_Type v_Request;
    if (p_USIM_Rmvd and pc_USIM_Rmv) {  // SIM needs to be inserted
      v_Request := INSERT_USIM;
    } else if (pc_SwitchOnOff) {        // switch on the UE
      v_Request := SWITCH_ON;
    } else if (pc_USIM_Rmv) {           // UE doesn't support switch off, so SIM needs to be inserted;  @sic R5-104083 sic@
      v_Request := INSERT_USIM;
    } else {                            // power on the UE
      v_Request := POWER_ON;
    }
    if ( p_CnfRequired) { 
      fl_UT_ApplyCommand(p_Port, cas_UT_Req(v_Request, CNF_REQUIRED, omit));
    } else {
      fl_UT_ApplyCommand(p_Port, cas_UT_Req(v_Request, NO_CNF_REQUIRED, omit));
    }
  }


Change 3 – Simplification of function ‘f_EUTRA_RmvFbdnPLMN’ 

	Function name
	f_EUTRA_RmvFbdnPLMN

	Reason for change
	This function takes too long (more than 5 minutes) but could be executed in a couple of seconds instead. 
UEs may trigger during this long time an Attach or TAU procedure when switching from manual to automatic PLMN search. This would cause the test case to FAIL even though the test body was already successfully PASSed.
Also it is not needed to pass parameters which can be computed from the cellId.

The UE will be attached again to ‘p_CellId_Detach’ when completing the procedure.

	Summary of change
	The function is completely redesigned and simplified as shown below.

	TTCN module
	EUTRA_NASCommonFunctions.ttcn

	MCC160 Comment
	Accepted


Before change

	  /*
   * @desc      Removal of forbidden PLMN
   * @param     p_CellId_Detach
   * @param     p_CellId_Removal
   * @param     p_CellPowerList
   * @param     p_PLMN_ID
   * @param     p_ForcedAttach
   * @status    APPROVED
   */
  function f_EUTRA_RmvFbdnPLMN ( CellId_Type p_CellId_Detach,
                                 CellId_Type p_CellId_Removal,
                                 template (value) CellPowerList_Type p_CellPowerList,
                                 NAS_PlmnId p_PLMN_ID,
                                 EUTRA_ATTACH_Type p_ForcedAttach ) runs on EUTRA_PTC // @sic R5-106675 sic@
  {
    //detach UE from last cell
    f_EUTRA_UE_Detach_SwitchOffUe ( p_CellId_Detach, RRC_CONNECTED, -, p_ForcedAttach); // @sic R5s110176 sic@
    //configure cells to make cell of forbidden PLMN serving cell
    f_EUTRA_SetCellPowerList ( p_CellPowerList );
    //switch on UE
    f_UT_SwitchOnUE (UT, false);
    //switch to automatic PLMN search
    f_UT_ManualPLMN_Select (UT, p_PLMN_ID);
    //registration on cell of forbidden PLMN
    f_EUTRA_AttachProcedure (p_CellId_Removal, "No Verdict", p_ForcedAttach); // @sic R5s110176 sic@
    //switch to automatic PLMN search
    f_UT_AutomaticPLMN_Select ( UT );
    //wait 300 seconds to let UE complete automatic PLMN search
    f_Delay(300.0);
  } // f_EUTRA_RmvFbdnPLMN


After change

	  /*
   * @desc      Removal of forbidden PLMN
   * @param     p_CellId_Detach
   * @param     p_CellId_Removal
   * @param     p_CellPowerList
   * @param     p_PLMN_ID
   * @param     p_ForcedAttach
   * @status    APPROVED
   */
  function f_EUTRA_RmvFbdnPLMN ( CellId_Type p_CellId_Detach,
                                 CellId_Type p_CellId_Removal,
                                 EUTRA_ATTACH_Type p_ForcedAttach ) runs on EUTRA_PTC // @sic R5-106675 sic@
  {
    var GutiParameters_Type v_Guti_Params;
    v_Guti_Params := f_EUTRA_CellInfo_GetGuti(p_CellId_Removal); 
    //detach UE from last cell
    f_EUTRA_UE_Detach_SwitchOffUe ( p_CellId_Detach, RRC_CONNECTED, -, p_ForcedAttach); // @sic R5s110176 sic@
    //Set new power levels to attach on second cell
    // f_EUTRA_SetCellPowerList ( p_CellPowerList );
    f_EUTRA_SetCellPower (p_CellId_Detach, tsc_NonSuitableOffCellRS_EPRE );
    f_EUTRA_SetCellPower (p_CellId_Removal, tsc_ServingCellRS_EPRE );
    //switch on UE
    f_UT_SwitchOnUE (UT, false, true ); 
    //switch to automatic PLMN search
    f_UT_ManualPLMN_Select (UT, f_Asn2Nas_PlmnId(v_Guti_Params.PLMN_Identity));
    //registration on cell of forbidden PLMN
    f_EUTRA_AttachProcedure (p_CellId_Removal, "No Verdict", p_ForcedAttach); 
    f_EUTRA_UE_Detach_SwitchOffUe ( p_CellId_Removal, RRC_IDLE); 
    f_EUTRA_SetCellPower (p_CellId_Removal, tsc_NonSuitableOffCellRS_EPRE ); 
    f_EUTRA_SetCellPower (p_CellId_Detach, tsc_ServingCellRS_EPRE );
    f_UT_SwitchOnUE (UT, false, true ); 
    f_UT_AutomaticPLMN_Select ( UT );
    f_EUTRA_AttachProcedure (p_CellId_Detach, "No Verdict", p_ForcedAttach); 
  } // f_EUTRA_RmvFbdnPLMN

